Matematická analýza

Limity a nekonečné součiny

V Maple není třeba rozlišovat mezi limitou funkce a posloupnosti takže následující limita může být stejně tak dobře limitou funkce i posloupnosti. Vše záleží na naší představě o proměnné n (tato poznámka platí i pro sumy a nekonečné součiny) :

```
> Limit((n^3+1)/(n^4+n^2+n), n=infinity)=limit((n^3+1)/(n^4+n^2+n), n=infinity);
```

$$\lim_{n \to \infty} \frac{n^3+1}{n^4+n^2+n} = 0$$

Samozřejmě můžeme meze snadno změnit na jakouko-li konstantu:

```
> Limit((1-cos(x))/x^2, x=0)=limit((1-cos(x))/x^2, x=0);
```

$$\lim_{x \to 0} \frac{1-\cos(x)}{x^2} = \frac{1}{2}$$

Tento výsledek Maple vrátí v případě neexistence limity:

```
> Limit(1/x,x=0)=limit(1/x,x=0);
```

$$\lim_{x \to 0} \frac{1}{x} = \mathit{undefined}$$

Někdy se můžeme setkat s nečekanými problémy:

```
> Limit(cos(sqrt(x+1))-cos(sqrt(x-1)),x=infinity)=limit(cos(sqrt(x+1))-cos(sqrt(x-1)),x=infinity);
```

$$\lim_{x \to \infty} \cos(\sqrt{x+1}) - \cos(\sqrt{x-1}) = -2 .. 2$$

Ale po jednoduché úpravě součtovým vzorcem cos(x) už je jasno:

```
> Limit(-2*sin((sqrt(x+1)+sqrt(x-1))/2)*sin((sqrt(x+1)-sqrt(x-1))/2),x=infinity)=limit(-2*sin((sqrt(x+1)+sqrt(x-1))/2)*sin((sqrt(x+1)-sqrt(x-1))/2),x=infinity);
```

$$\lim_{x \to \infty} 2 \sin\left(\frac{\sqrt{x+1}}{2} + \frac{\sqrt{x-1}}{2}\right) \sin\left(-\frac{\sqrt{x+1}}{2} + \frac{\sqrt{x-1}}{2}\right) = 0$$

Tato chyba se zřídka stává při rozdílu dvou nekonečných výrazů (byť třeba v čitateli nebo jmenovateli zlomků). Pokoušíme se tedy převádět na jednoduché tvary. Proč nám předchozí výsledek vyšel tak divně? Snad to ozřejmí následující případ, kdy Maple také nedokáže limitu spočítat.

```
> Limit((-1)^n,n=infinity)=limit((-1)^n,n=infinity);
```

$$\lim_{n \to \infty} (-1)^n = -1 .. 1$$

Pokud Maple nedokáže limitu spočítat nebo zjistit její neexistenci, vrátí pouze interval, kde by se v případě své existence mohla požadovaná limita nacházet. V našem případě ovšem víme, že -1 a 1 jsou pouze hromadné body. Vlastní limitu tato posloupnost (resp. fce) nemá

Samozřejmě Maple umí počítat i limity jednostranné:

```
> Limit(1/x, x=0, left)=limit(1/x, x=0, left);
> Limit(1/x, x=0, right)=limit(1/x, x=0, right);
```

$$\lim_{x \to 0-} \frac{1}{x} = -\infty$$

$$\lim_{x \to 0+} \frac{1}{x} = \infty$$

Výhodou Maplu je, že poměrně snadno zvládá i těžší, ne zcela triviální limity.

```
> Limit(x*(sqrt(x^2+2*x)-2*sqrt(x^2+x)+x),x=infinity)=limit(x*(
  sqrt(x^2+2*x)-2*sqrt(x^2+x)+x),x=infinity);
```

$$\lim_{x \to \infty} x\left(\sqrt{x^2+2x} - 2\sqrt{x^2+x} + x\right) = \frac{-1}{4}$$

Pomocí Maplu lze s úspěchem upočítat i nekonečné součiny :

```
> Product(1-1/(n+1)^2,n=1..infinity)=product(1-1/(n+1)^2,n=1..i
  nfinity);
```

$$\prod_{n=1}^{\infty} \left(1 - \frac{1}{(n+1)^2}\right) = \frac{1}{2}$$

Maple dokonce zná i Wallisovu formuli :

```
> Product((2*n)^2/((2*n-1)*(2*n+1)),
  n=1..infinity)=product((2*n)^2/((2*n-1)*(2*n+1)),
  n=1..infinity);
```

$$\prod_{n=1}^{\infty} \left(\frac{4\,n^2}{(2\,n-1)\,(2\,n+1)}\right) = \frac{\pi}{2}$$

Derivace a vyšetřování extrémů

S derivacemi je to jednoduché - stačí používat symbol parciální derivace (o této fci

Maplu viz. pro funkce více proměnných).

```
> Diff(sin(x)*x^7, x)=diff(sin(x)*x^7,x);
```

$$\frac{d}{dx}(\sin(x)\,x^7) = \cos(x)\,x^7 + 7\sin(x)\,x^6$$

A pro druhou derivaci podobně:

```
> Diff(sin(x)*x^7, x,x)=diff(sin(x)*x^7,x,x);
```

$$\frac{d^2}{dx^2}(\sin(x)\,x^7) = -\sin(x)\,x^7 + 14\cos(x)\,x^6 + 42\sin(x)\,x^5$$

Můžeme též použít diferenciální operátor a vyčíslit zderivovanou fci :

```
> D(sin^4);
> D(sin^4)(x);
```

$$4\cos\sin^3$$

$$4\cos(x)\sin(x)^3$$

A následně pro libovolnou derivaci (v našem případě pátou) :

```
> (D@@5)(sin^2)(x);
```

$$32\cos(x)\sin(x)$$

Samozřejmě konkrétní vyčíslení není žádný problém:

```
> (D@@5)(sin^2)(Pi);
```

$$0$$

Pro derivaci složené fce cos(f) můžeme použít :

```
> D(sin@f)(x);
```

$$\cos(\mathrm{f}(x))\,\mathrm{D}(f)(x)$$

Teď už k vyšetřování extrémů. Základní prostředek jsou funkce maximize a minimize.

```
> maximum:=maximize(sin(x)+cos(x), x);
> minimum:=minimize(sin(x)+cos(x), x);
```

$$maximum := \sqrt{2}$$

$$minimum := -\sqrt{2}$$

Pokud chceme hledat extrémy na celé reálné ose i s rozšířením do obou nekonečných hodnot, je potřeba používat speciální podmínku 'infinite'.

```
> minimize(exp(x),x);
```

$$0$$

Maple neposkytl žádný výsledek a správně tak odpověděl, že minimum není na reálné ose nabyto. Nicméně limita do $-\infty$ je 0 což je infimum celé funkce. Abychom získali tento výsledek použijeme podmínku 'infinite'.

```
> minimize(exp(x),x,'infinite');
```

$$0$$

Jak je vidět zašvindlovali jsme to úspěšně ;-).

Při vyšetřování průběhu nám často může napomoci i Maplovská fce solve, která počítá průniky dvou křivek :

```
> solve({sin(x)=y,y=0},[x,y]);
```

$$[[x=0, y=0]]$$

Tuto fci lze samozřejmě použít i v přídadě vyšetřování implicitně zadané fce např. Descartesova listu.

```
> solve({x^3+y^3-2*x*y=0, y=a*x},[x,y]);
```

$$\left[[x=0, y=0], [x=0, y=0], \left[x=\frac{2\,a}{1+a^3}, y=\frac{2\,a^2}{1+a^3}\right]\right]$$

Sumy konečné i nekonečné

Nejdřívě něco jednoduchého, spočteme třeba nějaký zajímavý výsledek ;-) :

```
> Sum(1/n,n=1..100)=sum(1/n,n=1..100);
```

$$\sum_{n=1}^{100} \frac{1}{n} = \frac{14466636279520351160221518043104131447711}{2788815009188499086581352357412492142272}$$

Sčítat konečné sumy ale není moc zajímavé, jejich sečtením se toho moc nedovíme, jak budeme demostrovat na následujícím příkladu. Naštěstí umí Maple sčítat i sumy nekonečné.

```
> Sum(1/n,n=1..1000000)=evalf(sum(1/n,n=1..1000000));
> Sum(1/n,n=1..infinity)=sum(1/n,n=1..infinity);
```

$$\sum_{n=1}^{1000000} \frac{1}{n} = 14.39272672$$

$$\sum_{n=1}^{\infty} \frac{1}{n} = \infty$$

Bohužel rozsah sum, které dokáže Maple s úspěchem spočítat není moc velký. Jsou to zejména sumy upočitatelné z teorie mocninných řad a Taylorových polynomů.

```
> Sum((2*n+1)/n!*x^(2*n),n=0..infinity)=sum((2*n+1)/n!*x^(2*n),
  n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{(2\,n+1)\,x^{(2\,n)}}{n!} = \mathbf{e}^{(x^2)}\,(1+2\,x^2)$$

Ale už na o málo těžším příkladě si Maple vyláme zuby a vypíše takovýto výsledek :

```
> Sum((n^2+1)/(2^n*n!)*x^n,n=0..infinity)=sum((n^2+1)/(2^n*n!)*
  x^n,n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{(n^2+1)\,x^n}{2^n\,n!} = \left(1+\frac{1}{2}\,x+\frac{1}{4}\,x^2\right)\mathbf{e}^{\left(\frac{x}{2}\right)}$$

Trigonometrické řady jsou pro Maple též moc velký výkon. :(

```
> Sum(cos(n*x)/n!, n=0..infinity)=sum(cos(n*x)/n!,
  n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{\cos(n\,x)}{n!} = \frac{1}{2}\,\mathbf{e}^{(\mathbf{e}^{(x\,I)})} + \frac{1}{2}\,\mathbf{e}^{(\mathbf{e}^{(-I\,x)})}$$

Natož aby si chudák poradil s Parsevalovou rovností.

```
> Sum(sin(n*alpha)/n^2,n=1..infinity)=sum(sin(n*alpha)/n^2,n=1.
  .infinity);
```

$$\sum_{n=1}^{\infty} \frac{\sin(n\,\alpha)}{n^2} = \frac{-1}{2}\,I\,(\mathrm{polylog}(2,\mathbf{e}^{(\alpha\,I)}) - \mathrm{polylog}(2,\mathbf{e}^{(-I\,\alpha)}))$$

Takže v oblasti sčítání nekonečných řad neprojeví Maple žádnou zvláštní sílu. Neumí ani pořádně zjistit jejich konvergenci, vše musíme dělat ručně. Na druhou stranu zná pár "profláknutých" výsledků. A ty pak umí většinou použít.

```
> Sum(1/n^2,n=1..infinity)=sum(1/n^2,n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2} = \frac{\pi^2}{6}$$

Taylorův polynom

Maple umí vytvořit Taylorův polynom v Peanově tvaru :

```
> sin(n*x)^2=taylor(sin(n*x)^2, x,12);
```

$$\sin(n\,x)^2 = n^2\,x^2 - \frac{n^4}{3}\,x^4 + \frac{2\,n^6}{45}\,x^6 - \frac{n^8}{315}\,x^8 + \frac{2\,n^{10}}{14175}\,x^{10} + \mathrm{O}(x^{12})$$

Samozřejmě nemusíme se omezovat pouze na Maclaurinovy polynomy a můžeme s chutí vytvářet Taylorovy polynomy libovolného stupně a v libovolném bodě (pokud tam samozřejmě existují)..

```
> ln(x)=taylor(ln(x),x=1,5);
```

$$\ln(x) = x - 1 - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + \mathrm{O}((x-1)^5)$$

V případě, že v daném bodě Taylorův polynom neexistuje můžeme použít fci series, která aproximuje fci řadou logaritmů a obecných mocnin :

```
> x^x=taylor(x^x,x=0,5);
> x^x=series(x^x,x=0,5);
Error, does not have a taylor expansion, try series()
```

$$x^x = 1 + \ln(x)\,x + \frac{1}{2}\ln(x)^2\,x^2 + \frac{1}{6}\ln(x)^3\,x^3 + \frac{1}{24}\ln(x)^4\,x^4 + \mathrm{O}(x^5)$$

Můžeme také sčítat nazpátek :

```
> Sum((-1)^n*x^(2*n+1)/(2*n+1)!,
  n=0..infinity)=sum((-1)^n*x^(2*n+1)/(2*n+1)!, n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{(-1)^n\,x^{(2\,n+1)}}{(2\,n+1)!} = \sin(x)$$

Integrály

Nejdříve se podíváme na neurčité integrály. Maple rychle spočítá i ty poměrně složité.

```
> int(1/(sqrt(1+exp(x))+sqrt(1-exp(x))), x);
```

$$-\frac{1}{4}\,\frac{1}{\sqrt{1-\mathbf{e}^x}+1} - \frac{1}{4}\,\frac{1}{-1+\sqrt{1-\mathbf{e}^x}} - \frac{1}{4}\ln(\sqrt{1-\mathbf{e}^x}+1) + \frac{1}{4}\ln(-1+\sqrt{1-\mathbf{e}^x})$$
$$- \frac{1}{4}\,\frac{1}{\sqrt{1+\mathbf{e}^x}+1} - \frac{1}{4}\,\frac{1}{-1+\sqrt{1+\mathbf{e}^x}} - \frac{1}{4}\ln(\sqrt{1+\mathbf{e}^x}+1) + \frac{1}{4}\ln(-1+\sqrt{1+\mathbf{e}^x})$$

Přechod k určitému integrálu je (alespoň v Maplu ;-)) jednoduchý. Jen doplníme meze.

```
> Int(1/sqrt(x^2+3*x+1), x=0..1)=int(1/sqrt(x^2+3*x+1),
  x=0..1);
```

$$\int_0^1 \frac{1}{\sqrt{x^2+3x+1}}\,dx = -\ln(5)+\ln(5+2\sqrt{5})$$

Výsledky z Maplu jsou sice vždy správné, ale často vypadají nepěkně anebo je dostaneme v dost odlišném tvaru, než čekáme.

```
> Int(1/(x*sqrt(-ln(x)^2+4*ln(x)-3)),x)=int(1/(x*sqrt(-ln(x)^2+
  4*ln(x)-3)),x);
```

$$\int \frac{1}{x\sqrt{-\ln(x)^2+4\ln(x)-3}}\,dx = \arcsin(\ln(x)-2)$$

To je sice pěkné, ale co když potřebujeme ověřit jestli :
$\int \frac{1}{x\sqrt{-\ln(x)^2+4\ln(x)-3}}\,dx = -2\,\mathrm{arctg}\left(\sqrt{\frac{3-\ln(x)}{\ln(x)-1}}\right)$ (jak jste po dlouhém počítání jistě spočetli sami ;-)). Občas pomůže odečtení výsledků - hned je vidět na čem jsme. Ale v tomto případe se ani po dlouhém hledání nepodařilo ztotožnit oba výsledky jinak než po ruční úpravě podle vzorců na arcsin a arctg.

Diferenciální rovnice

V této kapitole se Maple docela blýskne. Řeší se pomocí dsolve. Nejdříve něco jednoduchého.

```
> dsolve(diff(y(x),x)+2*x*y(x)=x*exp(-x^2), y(x));
```

$$y(x) = \left(\frac{x^2}{2} + \_C1\right) e^{(-x^2)}$$

Kde _C1 je samozřejmě příslušná konstanta. Maple zvládne i složitější věci se kterými se na cvičeních ani běžně nesetkáme:

```
> dsolve(diff(y(x),x)=(1+y(x)^2)*cos(x), y(x));
```

$$y(x) = \tan(\sin(x) + \_C1)$$

dsolve si poradí i s okrajovými podmínkami, jen přidáme další rovnici :

```
> dsolve([diff(y(x),x)=(1+y(x)^2)*cos(x),y(0)=sqrt(3)/3],
  y(x));
```

$$y(x) = \tan\left(\sin(x) + \frac{\pi}{6}\right)$$

Můžeme řešit i soustavy .

```
> soustava:=[diff(x(t),t)=x(t)-y(t)+z(t),diff(y(t),t)=x(t)+y(t)
  -z(t),diff(z(t),t)=2*z(t)-y(t)];
> dsolve(soustava,[x(t),y(t),z(t)]);
```

$$soustava := \left[\frac{d}{dt}\,x(t) = x(t) - y(t) + z(t), \frac{d}{dt}\,y(t) = x(t) + y(t) - z(t), \frac{d}{dt}\,z(t) = 2\,z(t) - y(t)\right]$$

$$\{x(t) = \_C1\,\mathbf{e}^{(2\,t)} + \_C2\,\mathbf{e}^{t} + \_C3\,\mathbf{e}^{t}\,t + \_C3\,\mathbf{e}^{t}, y(t) = \mathbf{e}^{t}\,(\_C2 + \_C3\,t - \_C3),$$
$$z(t) = \_C1\,\mathbf{e}^{(2\,t)} + \_C2\,\mathbf{e}^{t} + \_C3\,\mathbf{e}^{t}\,t\}$$

Maple si poradí i s parciálními diferenciálními rovnicemi :

```
> f:='f':
```

```
> pdsolve( diff(f(x,y),x,x)+5*diff(f(x,y),x,y)=3, f(x,y) );
```

$$f(x, y) = \_F1(y) + \_F2(y - 5\,x) + \frac{3\,x^2}{2}$$

Kde _F1 a _F2 jsou příslušné upřesňující funkce.

Pokud nechceme řešit rovnice klasickým zbůsobem, nebo postup takového řešení dokonce neexistuje, můžeme použít obsáhlou knihovnu numerických metod - viz Základy numerické matematiky .

Funkce více proměnných

Limity funkcí více proměnných

Je to opět jednoduché

```
> Limit(x+1/y, {x=0,y=infinity})=limit(x+1/y,
  {x=0,y=infinity});
```

$$\mathrm{Limit}\left(x + \frac{1}{y}, \{x = 0, y = \infty\}\right) = 0$$

A takhle to vypadá, když limita neexistuje :

```
> Limit((x^2+y)/(x+y), {x=0,y=0})=limit((x^2+y)/(x+y),
```

```
{x=0,y=0});
```

$$\text{Limit}\left(\frac{x^2+y}{x+y}, \{x=0, y=0\}\right) = \mathit{undefined}$$

Bohužel Maple si skoro vždy vyláme na těžších limitách zuby.

```
> limit(x*y/sqrt(x^2+y^2), {x=0,y=0});
```

$$\text{limit}\left(\frac{x\,y}{\sqrt{x^2+y^2}}, \{x=0, y=0\}\right)$$

Třeba na této, o níž víme, že je rovna 0.

Parciální derivace a totální diferenciál

To už vlastně umíme. Používáme starou známou fci diff.

```
> Diff(sin(x^2+y^2), x,y)=diff(sin(x^2+y^2), x,y);
```

$$\frac{\partial^2}{\partial y\,\partial x}\sin(x^2+y^2) = -4\sin(x^2+y^2)\,y\,x$$

Podobných výsledků lze dosáhnou i při použití diferenciálního operátoru D.

```
> f:= (x,y) -> sin(x+y)+y^2;
> D[1](f);
> D[2](f);
```

$$f := (x, y) \to \sin(x+y)+y^2$$

$$(x, y) \to \cos(x+y)$$

$$(x, y) \to \cos(x+y)+2\,y$$

Nic na tom není. ;-). Teď se podíváme jak si Maple poradí s gradientem. Před použitím je nutno načíst tento příklad z knihovny

```
> with(linalg,grad);
> grad(sin(x*y^2),vector([x,y]));
```

$$[\mathit{grad}]$$

$$[\cos(x\,y^2)\,y^2, 2\cos(x\,y^2)\,x\,y]$$

U vektorových funkcí Maple spočítá i rotaci:

```
> curl([sin(x*y^2*z),exp(x*y*z),z],vector([x,y,z]));
```

$$\text{curl}([\sin(x\,y^2\,z), \mathbf{e}^{(x\,y\,z)}, z], [x, y, z])$$

Taylorův polynom pro funkce více proměnných

Oproti funkcím jedné proměnné se změní málo - pouze použijeme místo funkce taylor funkci mtaylor. Ovšem nejdříve musíme načíst příslušnou knihovnu.

```
> readlib(mtaylor);
```

$$\mathbf{proc}()\ \ldots\ \mathbf{end\ proc}$$

Teď už se to podaří :

```
> mtaylor(sin(x+y), [x,y], 5);
```

$$x+y-\frac{1}{6}x^3-\frac{1}{2}y\,x^2-\frac{1}{2}x\,y^2-\frac{1}{6}y^3$$

Vytvoří Taylorův polynom v bodě [0,0]. Pro vytvoření v jiném bodě použijeme:

```
> mtaylor(sin(x+y), [x=Pi,y=Pi],4);
```

$$x-2\,\pi+y-\frac{(x-\pi)^3}{6}-\frac{(y-\pi)\,(x-\pi)^2}{2}-\frac{(x-\pi)\,(y-\pi)^2}{2}-\frac{(y-\pi)^3}{6}$$

Oproti Taylorovu polynomu fce jedné proměnné je v podstatě jedinná změna - nepracuje se se symbolem O(x).

Vyšetřování extrémů funkcí více proměnných

Na začátek něco jednoduchého. Můžeme funkce prostě maximalizovat jako v případě funkcí jedné proměnné

```
> with(Optimization);
```

$$[\mathit{ImportMPS}, \mathit{Interactive}, \mathit{LPSolve}, \mathit{LSSolve}, \mathit{Maximize}, \mathit{Minimize}, \mathit{NLPSolve}, \mathit{QPSolve}]$$

```
> extrema( x*y, {}, {x,y} );
```

$$\{0\}$$

Dále nás budou zajímat hlavně maxima fcí na množinách počítaná pomocí věty o Lagrangeových multiplikátorech. Používáme příkaz extrema obsažený v balíčku Student.

```
> with(student):
> extrema( a*x*y, x^2+y^2=1, {x,y} );
```

$$\{\max\left(-\frac{a}{2},\frac{a}{2}\right), \min\left(-\frac{a}{2},\frac{a}{2}\right)\}$$

První parametr je samozřejmě vyšetřovaná funkce, na druhém místě následují v závorce rovnice, které určují množinu vůči níž se počítají extrémy, na třetím místě se pak vloží proměnné.

Implicitní funkce

Maple provádí aplikaci věty o implicitní funkci (resp. funkcí) pomocí příkazu implicitdiff.

Takhle vypadá derivace implicitně zadané kružnice v bodě $\left[\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right]$:

```
> kruznice := y^2+x^2=1;
> implicitdiff(kruznice,y,x);
> derivace := (x,y) -> -x/y;
> derivace(sqrt(2)/2,sqrt(2)/2);
```

$$kruznice := x^2 + y^2 = 1$$

$$-\frac{x}{y}$$

$$derivace := (x, y) \rightarrow -\frac{x}{y}$$

$$-1$$

Situace pro více funkcí je obdobná. Z následujících implicitních funkcí vyjádříme t a z jako funkce proměnných x a y tj. spočítáme jejich parciální derivace:

```
> F1 := x^2+y^2-1/2*z^2-t^3 = 0;
> F2 := x + y + z - t -2 = 0;
> implicitdiff({F1,F2},{t(x,y),z(x,y)},{t,z},x,notation=Diff);
> implicitdiff({F1,F2},{t(x,y),z(x,y)},{t,z},y,notation=Diff);
```

$$F1 := x^2 + y^2 - \frac{z^2}{2} - t^3 = 0$$

$$F2 := x + y + z - t - 2 = 0$$

$$\{\left(\frac{\partial}{\partial x} t\right)_y = \frac{2\,x + z}{z + 3\,t^2}, \left(\frac{\partial}{\partial x} z\right)_y = \frac{2\,x - 3\,t^2}{z + 3\,t^2}\}$$

$$\{\left(\frac{\partial}{\partial y} t\right)_x = \frac{2\,y + z}{z + 3\,t^2}, \left(\frac{\partial}{\partial y} z\right)_x = \frac{2\,y - 3\,t^2}{z + 3\,t^2}\}$$

Teď už stačí jen čile dosazovat konkrétní hodnoty pro t(x,y) a z(x,y) - na tento úkol je nutno mít zadaný bod splňující rovnice F1 a F2 a další předpoklady věty (tj. nenulovost parciálních derivací). Například pro bod [1,-1,2,0] je tedy z = z(1,-1) = 2 a t = t(1,-1) = 0.

Fourierova transformace

Fourierova transformace se nalézá v balíčku inttrans.

```
> with(inttrans);
```

$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable]$

Teď už s chutí ;-) transformujeme :

```
> fourier(exp(-x^2/2),x,t);
```

$$e^{\left(-\frac{t^2}{2}\right)} \sqrt{2} \sqrt{\pi}$$

Tedy $F(\omega) = \int_{-\infty}^{\infty} f(t)\, e^{(-I\,\omega\,t)}\, dt$. Transformace se žádným způsobem nenormuje. K dispozici máme i Inverzní Fourierovu transformaci:

```
> invfourier(sqrt(Pi*2)*exp(-x^2/2), x, t);
```

$$e^{\left(-\frac{t^2}{2}\right)}$$

```
>
```

Ta v Maplu vypadá takhle : $F(\omega) = \frac{1 \int_{-\infty}^{\infty} f(t)\, e^{(I\,\omega\,t)}\, dt}{2\,\pi}$, tedy integrál je znormovaný tak, aby operace byla inverzní k Fourierově transformaci fourier.

Použitá literatura

Luděk Zajíček : Vybrané úlohy z matematické analýzy
B.P. Dědidovič : Sbornik zadač i upražnenij po matematičeskomu analizu
Jiří Kopáček a kol.: Příklady z matematiky pro fyziky II.
Vojtěch Jarník : Úvod do počtu diferenciálního (Diferenciální počet I.)